


Superficies

Representación gráfica de distintos tipos de superficies

■ Grafos

```
In[140]:= f[x_, y_] := Sin[x] * y^2 / 7  
X = {u, v, f[u, v]};  
ParametricPlot3D[Evaluate[X], {u, -5, 5}, {v, -5, 5}, PlotPoints -> 30]
```



■ Superficies Tubulares

Recordamos que, dada $\alpha(t)$ curva regular y $r < \min(R(t))$, se puede definir la superficie que envuelve de forma tubular a α como aquella que tiene por parametrización:

$$X(u, v) = \alpha(u) + r \cos(v) N(u) + r \sin(v) T(u).$$


Primero creamos los elementos que necesitamos de la curva α

```
In[143]:= Clear["Global`*"]  
norma[z_] := Sqrt[z.z];  
alpha = {3 * Cos[u], 3 * Sin[u], u};  
T1 = D[alpha, u] / norma[D[alpha, u]] // Simplify;  
B1 = Cross[D[alpha, u], D[alpha, {u, 2}]] /  
      norma[Cross[D[alpha, u], D[alpha, {u, 2}]]] // Simplify;  
N1 =  
Cross[  
  B1,  
  T1];
```

Y ahora la dibujamos

```
In[149]:= r = 6 / 8;
X = alpha + r * Cos[v] * N1 + r * Sin[v] * B1
ParametricPlot3D[Evaluate[X], {u, 0, 3 Pi},
{v, 0, 2 Pi}, ViewPoint -> {4.5, -7.5, 1.5}, PlotPoints -> 30]
```

$$\text{Out[150]= } \left\{ 3 \cos[u] - \frac{3}{4} \cos[u] \cos[v] + \frac{3 \sin[u] \sin[v]}{4 \sqrt{10}}, \right. \\ \left. 3 \sin[u] - \frac{3}{4} \cos[v] \sin[u] - \frac{3 \cos[u] \sin[v]}{4 \sqrt{10}}, u + \frac{9 \sin[v]}{4 \sqrt{10}} \right\}$$



■ Superficies de traslación

Partimos de dos curvas, α y β y buscamos sus puntos de corte.

```
In[152]:= Clear["Global`*"]
a = 2; b = 3;
alpha[u_] := {u, 0, u^2 / a};
beta[v_] := {0, v - 1, -(v - 1)^2 / b}
ss = Solve[alpha[u] == beta[v], {u, v}]
```

```
Out[156]= {{u -> 0, v -> 1}}
```


El punto de corte es $P = \alpha(0) = \beta(1) = (0, 0, 0)$.

La parametrización es:

$$X(u, v) = \alpha(u) + \beta(v) - P$$

```
In[157]:= P = alpha[u] /. ss[[1]];
X = alpha[u] + beta[v] - P
ParametricPlot3D[{alpha[u], beta[v]}, {u, -2, 2}, {v, -1, 3}]
ParametricPlot3D[Evaluate[X], {u, -2, 2}, {v, -1, 3}]
```


```
Out[158]=  $\left\{u, -1 + v, \frac{u^2}{2} - \frac{1}{3}(-1 + v)^2\right\}$ 
```



■ Superficies de revolución

Partimos de una función $y = f(z) > 0$ y la giramos en torno al eje OZ

```
In[161]:= Clear["Global`*"]
f[z_] := 2 + Cos[z] * Sin[2 z];
X = {Cos[u] * f[v], Sin[u] * f[v], v};
ParametricPlot3D[Evaluate[X], {u, 0, 2 Pi}, {v, 0, 8}]
```



■ Superficies regladas

■ Paraboloid hiperbólico

A partir de curva $\alpha(u)$, y dirección generatriz: $w(u)$ ya tenemos nuestra parametrización:

$$X(u,v) = \alpha(u) + v w(u)$$


Vamos a hacer la del paraboloid hiperbólico (silla de montar) $z=kxy$: Obsevamos que, efectivamente, es doblemente reglada.

```

In[165]:= k = 1 / 10; alpha = {u, 0, 0}; w = {0, 1 / k, u};
X = alpha + v w;
Print["Ecuación reglada del paraboloides hiperbólico: x(u,v) = ", X]
ParametricPlot3D[Evaluate[X], {u, -5, 5}, {v, -5, 5}, AxesLabel -> {x, y, z}]

```

Ecuación reglada del paraboloides hiperbólico: $x(u,v) = \{u, 10 v, u v\}$




■ Cono

A partir de un vértice P y una curva $\alpha(u)$, tomamos como dirección generatriz: $w(u) = \alpha(u) - P$ yya tenemos nuestra parametrización:

$X(u,v) = \alpha(u) + v w(u)$ (ó también $X_2(u,v) = p + v w(u)$) (véase).

```
In[169]:= Clear["Global`*"]  
P = {1, 2, 3};  
alpha = {Cos[u], Sin[2 u] Cos[u], 1 - Cos[u] - Sin[u]};  
w = alpha - P;  
X = P + v w;  
Print["Ecuación de la superficie cónica x(u,v) = ", X]  
ParametricPlot3D[Evaluate[X], {u, 0, 2 Pi}, {v, -2, 4}, AxesLabel -> {x, y, z}]
```

Ecuación de la superficie cónica $x(u,v)$ =
{1 + v (-1 + Cos[u]), 2 + v (-2 + Cos[u] Sin[2 u]), 3 + v (-2 - Cos[u] - Sin[u])}



■ Hiperboloide

Dados $a, b, c > 0$, tomamos el hiperboloide de ecuación: $\frac{x^2}{a^2} + \frac{y^2}{b^2} - \frac{z^2}{c^2} = 1$.

Su generatriz, es la elipse que resulta al cortar con el plano ($z = 0$), esto es $\alpha(u) = (a \cos(u), b \sen(u), 0)$ y su generatriz $w(u) = \alpha'(u) + (0, 0, c) = (-a \sen(u), b \cos(u), c)$.

La parametrización queda:

$$X(u,v) = \alpha(u) + v w(u) = (a [\cos(u) - v \sen(u)], b [\sen(u) + v \cos(u)], c v)$$

```

In[176]:= Clear["Global`*"]
           {a, b, c} = {1, 2, 3};
           alpha = {a Cos[u], b Sin[u], 0};
           w = D[alpha, u] + {0, 0, c};
           X = alpha + v w
           ParametricPlot3D[Evaluate[X], {u, 0, 2 Pi}, {v, -3, 3}, AxesLabel -> {x, y, z}]

```

```


Out[180]= {Cos[u] - v Sin[u], 2 v Cos[u] + 2 Sin[u], 3 v}

```

```

Out[181]=

```



Geometría de superficies (primera y segunda forma fundamental)

■ Primera forma fundamental a partir de una parametrización

Dada $S = \{X(u,v), (u,v) \in U \subset \mathbb{R}^2\}$ la primera forma fundamental (sus componentes respecto de X_u y X_v) son

$$E = |X_u|^2$$

$$F = \langle X_u, X_v \rangle$$

$$G = |X_v|^2$$

Lo escribmos sobre un ejemplo, el **toro** (hecho en clase):

```
In[182]:= Clear["Global`*"]  
X = {(a + r * Cos[u]) * Cos[v], (a + r * Cos[u]) * Sin[v], r * Sin[u]};  
EE = D[X, u].D[X, u] // Simplify;  
FF = D[X, u].D[X, v];  
GG = D[X, v].D[X, v] // Simplify;  
Print[  
  "La primera forma fundamental del toro en la parametrización X(u,v) es: \n I(u,v) = ",  
  {{EE, FF}, {FF, GG}} // MatrixForm]
```

La primera forma fundamental del toro en la parametrización $X(u,v)$ es:

$$I(u,v) = \begin{pmatrix} r^2 & 0 \\ 0 & (a + r \cos[u])^2 \end{pmatrix}$$

■ Área de una superficie parametrizada

■ Área usando la fórmula base: $A = \int |X_u \times X_v| d(uv)$


Partimos de una superficie que se construye mediante combinaciones convexas del segmento \overline{ab} y el \overline{cd}


```

In[188]:= Clear["Global`*"]
a = {2, 0, 4};
b = {2, 4, 3};
c = {0, 4, 3};
d = {0, 0, 2};
segmento1 = a (1 - u) + b u;
segmento2 = c u + (1 - u) d;
X = Simplify[(1 - v) * segmento1 + v * segmento2];
Print["Parametrización del la superficie x(u,v)=", X]
ParametricPlot3D[Evaluate[X], {u, 0, 1}, {v, 0, 1}]

Parametrización del la superficie x(u,v)={2 - 2 v, 4 u, 4 - 2 v + u (-1 + 2 v)}

```



Calculamos el área desde la definición: $A = \int |X_u \times X_v| d(uv)$

```

In[198]:= n = Cross[D[X, u], D[X, v]];
area = NIntegrate[Evaluate[Sqrt[n.n]], {u, 0, 1}, {v, 0, 1}]

```

Out[199]= 9.25525

■ Área usando la primera forma fundamental: $A = \int \sqrt{EG - F^2} d(uv)$

Usamos el ejemplo de una esfera de radio R en coordenadas esféricas (ejercicio S2).


Determinamos la primera forma fundamental y calculamos el Área aplicando la fórmula de clase a partir de la

parametrización: $A = \int \sqrt{EG - F^2} d(uv)$

```

In[200]:= Clear["Global`*"]
X = {r Cos[v] Sin[u], r Sin[v] Sin[u], r Cos[u]};
(*la dibujamos con r=2*)
ParametricPlot3D[Evaluate[X] /. {r -> 2}, {u, 0, Pi}, {v, 0, 2 Pi}]
EE = D[X, u].D[X, u];
FF = D[X, u].D[X, v];
GG = D[X, v].D[X, v];
area = Integrate[Evaluate[Sqrt[EE * GG - FF^2]], {u, 0, Pi}, {v, 0, 2 Pi}]

```



Out[206]= $4\pi\sqrt{r^4}$

■ Área en el caso de un grafo

Dada la superficie (Véase ejercicio S27: Bóveda de Viviani) cuya ecuación implícita es la de una esfera de radio $2r$
 $x^2 + y^2 + z^2 = 4r^2$


pero restringida a $\{z > 0\}$ y conjunto de valores en el disco: $U = \{(x, y) \text{ tal que } (x-r)^2 + y^2 < r^2\}$.

Lo nuevo aquí es parametrizar U : ya que

$$x \in (0, 2r) \quad y \in \left(-\sqrt{r^2 - (x-r)^2}, \sqrt{r^2 - (x-r)^2} \right),$$

Vamos a representarla para $r = 1$ como grafo (despejando la z) y a calcular su área.

```
In[207]:= Clear["Global`*"]
f[x_, y_] := Sqrt[4 r^2 - x^2 - y^2];
X = {x, y, f[x, y]};
r = 2;
ParametricPlot3D[Evaluate[X], {x, 0, 2 r},
  {y, -Sqrt[r^2 - (x - r)^2], Sqrt[r^2 - (x - r)^2]}]
```



Y calculamos el Área aplicando la fórmula de clase a partir de la parametrización: $A = \int \sqrt{1 + |\nabla f|^2} d(xy)$

```
In[212]:= r = 2;
integrando = Sqrt[1 + D[f[x, y], x]^2 + D[f[x, y], y]^2] // Simplify;
area = NIntegrate[Evaluate[integrando],
  {x, 0, 2 r}, {y, -Sqrt[r^2 - (x - r)^2], Sqrt[r^2 - (x - r)^2]}]
```

Out[214]= 18.2655

■ Aplicación de Gauss y segunda forma fundamental a partir de una parametrización

Dada $S = \{X(u,v), (u,v) \in U \subset \mathbb{R}^2\}$ recordamos que la aplicación de Gauss es la que asigna el vector normal:

$$N = \frac{X_u \times X_v}{|X_u \times X_v|}$$

y las componentes de II son (respecto de X_u, X_v)

$$e = \langle N, X_{uu} \rangle$$

$$f = \langle N, X_{uv} \rangle$$

$$g = \langle N, X_{vv} \rangle$$

mientras que la matriz de para calcular las curvaturas principales, respecto de la base $B = \{X_u, X_v\}$, es

$$\begin{pmatrix} E & F \\ F & G \end{pmatrix}^{-1} \begin{pmatrix} e & f \\ f & g \end{pmatrix}$$

Lo aplicamos sobre un **paraboloide hiperbólico** (silla de montar) vista como grafo $z = x^2 - y^2$

```
In[215]:= Clear["Global`*"]
X = {u, v, u^2 - v^2};
n = Cross[D[X, u], D[X, v]];
NN = n / Sqrt[n.n];
Print["Función de Gauss N(u,v) = ", NN]

ee = NN.D[X, {u, 2}] // Simplify;
ff = NN.D[X, u, v];
gg = NN.D[X, {v, 2}] // Simplify;
matrizII = {{ee, ff}, {ff, gg}};
Print["Matriz de II(u,v) = ", matrizII // MatrixForm]
```

$$\text{Función de Gauss } N(u,v) = \left\{ -\frac{2u}{\sqrt{1+4u^2+4v^2}}, \frac{2v}{\sqrt{1+4u^2+4v^2}}, \frac{1}{\sqrt{1+4u^2+4v^2}} \right\}$$

$$\text{Matriz de II}(u,v) = \begin{pmatrix} \frac{2}{\sqrt{1+4u^2+4v^2}} & 0 \\ 0 & -\frac{2}{\sqrt{1+4u^2+4v^2}} \end{pmatrix}$$

Las otras matrices son

```
In[225]:= EE = D[X, u].D[X, u] // Simplify;
FF = D[X, u].D[X, v];
GG = D[X, v].D[X, v] // Simplify;
matrizI = {{EE, FF}, {FF, GG}};
matrizcurvaturas = Inverse[matrizI].matrizII;
Print["Matriz de I(u,v) = ", matrizI // MatrixForm]
Print["Matriz de las curvaturas = ", matrizcurvaturas // MatrixForm]
```

$$\text{Matriz de I}(u,v) = \begin{pmatrix} 1+4u^2 & -4uv \\ -4uv & 1+4v^2 \end{pmatrix}$$

$$\text{Matriz de las curvaturas} = \begin{pmatrix} \frac{2(1+4v^2)}{(1+4u^2+4v^2)^{3/2}} & -\frac{8uv}{(1+4u^2+4v^2)^{3/2}} \\ \frac{8uv}{(1+4u^2+4v^2)^{3/2}} & -\frac{2(1+4u^2)}{(1+4u^2+4v^2)^{3/2}} \end{pmatrix}$$

■ Curvatura de Gauss, curvatura media, curvaturas principales

Una vez que tenemos la matriz de las curvaturas (apartado anterior), recordamos las definiciones

$-k_1$ y k_2 son los autovalores de la matriz de curvaturas

$-K$ = curvatura de Gauss = det (matrizcurvaturas) = $k_1 k_2$,

$-H$ = curvatura media = $\frac{1}{2}$ traza (matrizcurvaturas) = $\frac{k_1+k_2}{2}$.

In[232]:=

```
{k1, k2} = Eigenvalues[matrizcurvaturas] // Simplify;
KK = Det[matrizcurvaturas] // Simplify;
HH = (matrizcurvaturas[[1, 1]] + matrizcurvaturas[[2, 2]]) / 2 // Simplify;

Print["Curvaturas principales: k1 = ", k1, ", y k2 = ", k2 ]
Print["Curvatura de Gauss K = ", KK ]
Print["Curvatura media = ", HH ]
```

Curvaturas principales: k1 =

$$-\frac{1}{(1+4u^2+4v^2)^{7/2}} 2 \left(2u^2 + 16u^4 + 32u^6 - 2v^2 + 32u^4v^2 - 16v^4 - 32u^2v^4 - 32v^6 + \sqrt{(1+4u^2+4v^2)^4 (4u^4 + u^2(4-8v^2) + (1+2v^2)^2)} \right), \text{ y } k_2 =$$

$$\frac{1}{(1+4u^2+4v^2)^{7/2}} 2 \left(-2u^2 - 16u^4 - 32u^6 + 2v^2 - 32u^4v^2 + 16v^4 + 32u^2v^4 + 32v^6 + \sqrt{(1+4u^2+4v^2)^4 (4u^4 + u^2(4-8v^2) + (1+2v^2)^2)} \right)$$

$$\text{Curvatura de Gauss } K = -\frac{4}{(1+4u^2+4v^2)^2}$$

$$\text{Curvatura media} = -\frac{4(u^2 - v^2)}{(1+4u^2+4v^2)^{3/2}}$$

■ Clasificación de puntos

Recordamos que K caracteriza los puntos

$-K < 0$ punto hiperbólico (dos direcciones perpendiculares con curvaturas opuestas, tipo silla de montar)

$-K > 0$ punto elíptico (en todas las direcciones, las curvas están "hacia el mismo lado", tipo elipsoide)

$-K = 0$ punto parabólico (tipo cilindro) que puede ser plano si $H = 0$ también.

Se dice umbilical cuando $k_1 = k_2$ (elíptico pero con igual curvatura en todas direcciones, tipo esfera)

In[238]:=

```
P = {u -> 1, v -> 1};
Print["Punto P = ", X /. P]
Print["Curvatura de Gauss K en P = ", KK /. P]
Print["Curvatura media H en P = ", HH /. P]
Print["curvatura principal k1 en P = ", k1 /. P]
Print["curvatura principal k2 en P = ", k2 /. P]
```

Punto $P = \{1, 1, 0\}$

Curvatura de Gauss K en $P = -\frac{4}{81}$

Curvatura media H en $P = 0$

curvatura principal k_1 en $P = -\frac{2}{9}$

curvatura principal k_2 en $P = \frac{2}{9}$

En este caso, $K < 0$, tenemos un punto no umbilical (hiperbólico), por lo que podemos calcular sus direcciones principales:

■ Direcciones principales, indicatriz de Dupin, dirección asintótica

Las direcciones principales son:

```
In[244]:= {v1, v2} = Eigenvectors[matrizcurvaturas /. P];
Print["Dirección principal v1 en P respecto de la base {Xu, Xv}= ", v1]
Print["Dirección principal v1 en P = ", (v1[[1]] * D[X, u] /. P) + (v1[[2]] * D[X, v] /. P)]
Print["Dirección principal v2 en P respecto de la base {Xu, Xv}= ", v2]
Print["Dirección principal v2 en P = ", (v2[[1]] * D[X, u] /. P) + (v2[[2]] * D[X, v] /. P)]
```

Dirección principal v_1 en P respecto de la base $\{X_u, X_v\} = \left\{\frac{1}{2}, 1\right\}$

Dirección principal v_1 en $P = \left\{\frac{1}{2}, 1, -1\right\}$

Dirección principal v_2 en P respecto de la base $\{X_u, X_v\} = \{2, 1\}$

Dirección principal v_2 en $P = \{2, 1, 2\}$

La indicatriz de Dupin son las cónicas en $T_p S$ de coordenadas (a, b) cumpliendo:

$$k_1 a^2 + k_2 b^2 = 1, k_1 a^2 + k_2 b^2 = -1$$

Dejamos como ejercicio el representarla.

```
In[249]:= a^2 (k1 /. P) + b^2 (k2 /. P) == -1
a^2 (k1 /. P) + b^2 (k2 /. P) == 1
```

```
Out[249]= -\frac{2 a^2}{9} + \frac{2 b^2}{9} == -1
```

```
Out[250]= -\frac{2 a^2}{9} + \frac{2 b^2}{9} == 1
```

En este caso, $K < 0$, y las direcciones asintóticas (vectores con curvatura normal cero) tienen sentido precisamente en puntos hiperbólicos (o parabólicos).

En términos de la segunda forma es $\Pi_p(w) = 0$ (en coordenadas respecto de $\{X_u, X_v\}$) será:

$$\begin{pmatrix} a & b \end{pmatrix} \begin{pmatrix} e & f \\ f & g \end{pmatrix} \begin{pmatrix} a \\ b \end{pmatrix} = 0.$$

```
In[251]:= matrizII = {{ee, ff}, {ff, gg}} /. P;
asin = Solve[{a, b}.matrizII.{a, b} == 0, {a, b}];
{a, b} /. asin[[1]]
{a, b} /. asin[[2]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. >>

```
Out[253]= {-b, b}
```

```
Out[254]= {b, b}
```

En este caso, las direcciones asintóticas son dos (las asíntotas de la indicatriz de Dupin), en las direcciones $w_1 = (1, -1)$ y $w_2 = (1, 1)$. En coordenadas cartesianas, serían (dejamos libre el valor de a)

```
In[255]:= Print["Dirección asintótica w1 en P = ", a * (D[X, u] /. P) - (a * D[X, v] /. P)]
Print["Dirección asintótica w2 en P = ", a * (D[X, u] /. P) + (a * D[X, v] /. P)]
```

Dirección asintótica w_1 en $P = \{a, -a, 4a\}$

Dirección asintótica w_2 en $P = \{a, a, 0\}$

■ Ejercicio 1

a) Determina una parametrización para la superficie que resulta de rotar la curva

$$\alpha(t) = (2 + \cos(t), \sin(t), t)$$


en torno al eje OZ.

b) Dibújala

c) Verifica que una de sus curvaturas principales no cambia de signo, y la otra sí.

d) Determina un punto elíptico, uno parabólico, uno hiperbólico y uno umbilical.

■ Posible solución :



Punto elíptico $P = \{\sqrt{5}, 0, \frac{\pi}{2}\}$ con curvaturas principales $-\frac{1}{3}$ y $-\frac{4}{27}$

Punto hiperbólico $P = \{1, 0, \pi\}$ con curvaturas principales -1 y 2

Punto parabólico $P = \{\sqrt{3}, 0, -\frac{2\pi}{3}\}$ con curvaturas principales $-\frac{1}{\sqrt{6}}$ y 0

Punto umbilical $P = \{\sqrt{7}, 0, -\frac{\pi}{3}\}$ con curvaturas principales $-\frac{1}{\sqrt{10}}$ y $-\frac{1}{\sqrt{10}}$


■ Solución :

■ Ejercicio 2

Crea una superficie (a través de su parametrización) de modo que su representación gráfica se asemeje lo máximo posible a la cubierta de las Bodegas Ysios (ver foto abajo y úsese cualquier fuente de información que se considere necesaria). Representala gráficamente para corroborar la similitud.



■ Posible solución :



■ Solución :